

Mit Excel-Uhrzeit rechnen

Inhalt

So können Sie mit Uhrzeiten in Excel rechnen	2
Differenz zwischen zwei Uhrzeiten berechnen	11
Bearbeitungsdauer berechnen	19
Zeitangaben in Excel auf die Viertelstunde runden	25
Datum- und Zeitformate in Excel selbst definieren	30
Negative Uhrzeiten in Excel darstellen	36
Negative Uhrzeiten im 1900-Datumsystem darstellen	41
Empfehlungen aus dem Management-Handbuch	46

So können Sie mit Uhrzeiten in Excel rechnen

Wenn Sie Uhrzeiten im Format Stunden:Minuten in Excel erfassen, müssen Sie bei Berechnungen einige Besonderheiten beachten. Erfahren Sie, wie Sie Zellen bei der Addition von Uhrzeiten formatieren und wie Sie Formeln bei der Multiplikation mit Uhrzeiten korrigieren müssen.

Zuletzt geändert am 11.03.2026



In Excel können Sie Uhrzeiten direkt in Zellen erfassen. Mit diesen Uhrzeiten können Sie auch rechnen und Uhrzeiten summieren, subtrahieren oder mit einem anderen Wert multiplizieren.

Allerdings gibt es hierbei einige Besonderheiten zu beachten, die im Folgenden erläutert sind.

Addieren von Uhrzeiten

In der folgenden Abbildung soll die Wochenarbeitszeit eines Arbeitnehmers erfasst werden. Die geleisteten Stunden werden hierbei im Bereich B2:B6 eingetragen.

An den ersten beiden Tagen sind die geleisteten Stunden als Uhrzeit (Stunden:Minuten) bereits erfasst. Diese beiden erfassten Uhrzeiten sollen nun in der Zelle B7 als Summe angezeigt werden.

	A	B	C	D	E
1		Zeit			
2	Tag 1	08:00			
3	Tag 2	07:45			
4	Tag 3				
5	Tag 4				
6	Tag 5				
7	Summe				
8					

Erfasste Uhrzeiten, die summiert werden sollen

Die Summierung von Uhrzeiten können Sie grundsätzlich wie bei normalen Zahlen in Excel vornehmen. Erfassen Sie einfach in der Zelle B7 die Formel:

=SUMME(B2:B6)

Bestätigen Sie die Eingabe, indem Sie die **ENTER-Taste** drücken.

	A	B	C	D	E
1		Zeit			
2	Tag 1	08:00			
3	Tag 2	07:45			
4	Tag 3				
5	Tag 4				
6	Tag 5				
7	Summe	15:45			
8					

Einfache Summierung von Uhrzeiten

Addition von Uhrzeiten mit mehr als 24 Stunden

Nun werden für die restlichen Wochentage die geleisteten Stunden ebenfalls erfasst. Die Uhrzeiten ergeben eine Summe von 40 Stunden. In der Zelle B7 wird aber lediglich eine Summe von 16 Stunden angezeigt.

	A	B	C	D	E
1		Zeit			
2	Tag 1	08:00			
3	Tag 2	07:45			
4	Tag 3	08:00			
5	Tag 4	09:00			
6	Tag 5	07:15			
7	Summe	16:00			
8					

Summe wird falsch angezeigt bei Stundenzahl größer als 24

Die Anzeige der falschen Stundensumme liegt daran, dass Excel bei der Erfassung von Uhrzeiten grundsätzlich ein Zahlenformat verwendet, das nur die Anzeige 24 Stunden zulässt.

Sie bekommen daher in diesem Beispiel nur die Stunden angezeigt, welche die Zahl von 24 Stunden überschreiten: 40 Stunden - 24 Stunden = 16 Stunden


Dieses Problem können Sie beseitigen, indem Sie das Zahlenformat in der Summenzelle B7 anpassen.

Markieren Sie hierzu die Zelle B7 und drücken Sie anschließend die Tastenkombination **Strg + 1**. Es öffnet sich das Dialogfeld **Zellen formatieren**.

Aktivieren Sie das Register **Zahlen**.


Passen Sie jetzt das Zahlenformat für die Uhrzeit an, indem Sie die Formatsymbole für die Stunden **hh** in eckige Klammern setzen. Durch diese Anpassung im Zahlenformat ist die Stundenanzeige nicht mehr auf 24 Stunden begrenzt.

Bestätigen Sie Ihre Einstellung, indem Sie das Dialogfeld **Zellen formatieren** durch einen Klick auf **OK** schließen.



Zahlenformat für Uhrzeit anpassen

Durch die Anpassung des Zahlenformats werden die summierten Stunden jetzt richtig angezeigt.



	A	B	C	D	E
1		Zeit			
2	Tag 1	08:00			
3	Tag 2	07:45			
4	Tag 3	08:00			
5	Tag 4	09:00			
6	Tag 5	07:15			
7	Summe	40:00			
8					

Stunden werden richtig angezeigt mit dem angepassten Uhrzeit-Zahlenformat

Multiplikation mit Uhrzeiten

Eine weitere Besonderheit ist zu beachten, wenn Sie mit Uhrzeiten multiplizieren.

Beispiel: Sie wollen für eine Angebotskalkulation den Preis berechnen. Dazu multiplizieren Sie den geplanten Arbeitsaufwand in Stunden und im Format Stunden:Minuten (hh:mm) mit einem Stundensatz in Euro.

Im Beispiel der folgenden Abbildung beträgt der Stundensatz 20,00 EUR. In jeder Zeile soll der monetäre Aufwand entsprechend berechnet und dargestellt werden. Hierzu wird der Stundensatz in der Spalte C eingetragen.

Anschließend multiplizieren Sie in der Spalte D die Stundenzahl (Spalte B) mit dem Stundensatz (Spalte C). Erfassen Sie hierzu in der Zelle D2 die Formel:

=C2*B2

Als Ergebnis erhalten Sie den Wert 6,67 angezeigt. Obwohl die Formel auf den ersten Blick richtig ist, ist der berechnete Wert falsch, da 8 Stunden multipliziert mit 20,00 Euro einen Wert von 160,00 Euro ergeben müsste.

D2 X ✓ <i>f_x</i> =C2*B2				
	A	B	C	D
1		Zeit	Satz	Wert
2	Tag 1	08:00	20,00	6,67
3	Tag 2	07:45	20,00	
4	Tag 3	08:00	20,00	
5	Tag 4	09:00	20,00	
6	Tag 5	07:15	20,00	
7	Summe	40:00	20,00	
8				


Einfache Multiplikation mit Uhrzeiten liefert ein falsches Ergebnis

Die falsche Berechnung ergibt sich aus der Art und Weise, wie Excel intern Datumswerte und Zeiten interpretiert. Grundsätzlich sind in Excel **Uhrzeiten** eigentlich **Dezimalzahlen**, die nur in einem anderen Format dargestellt werden. Es gilt:

- 24 Stunden entsprechen der Zahl 1 (24/24)
- 12 Stunden entsprechen der Zahl 0,5 (12/24)
- 6 Stunden entsprechen der Zahl 0,25 (6/24)
- 1 Stunde entspricht 0,04 (1/24)
- 30 Minuten entsprechen 0,02 (0,5/24) oder (30/24/60) etc.

Die Stundenzahl 8 in der Zelle B2 entspricht somit dem Wert 0,33 ($8/24 = 0,33$). Dies können Sie überprüfen, indem Sie die erfassten Stunden in der Spalte B mit dem Zahlenformat **Zahl** anstelle des Zahlenformats **Uhrzeit** anzeigen lassen:

- Markieren Sie den Bereich **B2:B7**
- Öffnen Sie das Dialogfeld **Zellen formatieren**, indem Sie die Tastenkombination **Strg + 1** drücken.
- Wählen Sie das Register **Zahlen** aus.
- Wählen Sie unter der Kategorie **Zahl** ein Format mit 2 Dezimalstellen aus.
- Bestätigen Sie Ihre Einstellung, indem Sie das Dialogfeld durch einen Klick auf **OK** schließen.



Zahlenformat von Uhrzeit in Zahl ändern

Durch die Anpassung des Zahlenformats bekommen Sie jetzt die „tatsächlichen“ Zahlen im Bereich **B2:B7** dargestellt, mit denen Excel in der Spalte D auch rechnet. In der Zelle B2 steht der Wert 0,33. Dieser multipliziert mit dem Stundensatz 20 ergibt den Wert 6,67.

B2					
	A	B	C	D	E
1		Zeit	Satz	Wert	
2	Tag 1	0,33	20,00	6,67	
3	Tag 2	0,32	20,00		
4	Tag 3	0,33	20,00		
5	Tag 4	0,38	20,00		
6	Tag 5	0,30	20,00		
7	Summe	1,67	20,00		
8					

Uhrzeiten im Zahlenformat Zahl

Um die Problematik zu lösen, müssen Sie daher die Formel in der Zelle D2 anpassen. Da eine Stunde in Excel 1/24 darstellt, müssen Sie das ursprüngliche **Ergebnis noch mit 24 multiplizieren**. Modifizieren Sie daher die Formel in der Zelle D2 wie folgt:

=C2*B2*24

Kopieren Sie anschließend die Formel in den Bereich D3:D7 nach unten.

D2				
	A	B	C	D
1		Zeit	Satz	Wert
2	Tag 1	08:00	20,00	160,00
3	Tag 2	07:45	20,00	155,00
4	Tag 3	08:00	20,00	160,00
5	Tag 4	09:00	20,00	180,00
6	Tag 5	07:15	20,00	145,00
7	Summe	40:00	20,00	800,00
8				

Durch Multiplikation mit 24 stimmt das Ergebnis der Multiplikation mit Uhrzeiten

Merke**Uhrzeiten in Excel müssen umgerechnet werden**

Wenn Sie Berechnungen in Excel mit Uhrzeiten vornehmen, beachten Sie, dass Excel die dargestellten Uhrzeiten intern anders verarbeitet.

Differenz zwischen zwei Uhrzeiten berechnen

So berechnen Sie die Zeitdifferenz – inklusive Übernacht-Berechnung mit Excel. Die Zahl der Stunden und Minuten zwischen einer Start- und Endzeit lässt sich durch Subtraktion berechnen. Liegt die Endzeit am nächsten Tag, muss die Formel angepasst werden.

Zuletzt geändert am 11.03.2026



Die Berechnung der Stunden zwischen zwei Uhrzeiten ist in Excel eine häufige Aufgabe – sei es für die Arbeitszeiterfassung, im Projektmanagement oder für persönliche Zeitpläne. Doch wie berechnen Sie **Zeitspannen, die über Mitternacht hinausgehen?**

Eine einfache Methode, um die Stunden zwischen zwei Zeiten zu berechnen, ist die Subtraktion der Startzeit von der Endzeit. Bei Zeitspannen, die über Mitternacht hinausgehen, braucht die Formel jedoch eine Anpassung.

In diesem Beitrag erfahren Sie, wie Sie sowohl einfache Zeitspannen als auch solche über Mitternacht korrekt in Excel berechnen.

Beispiel: Stundendifferenz berechnen innerhalb eines Tages

Angenommen, Sie möchten die Stunden zwischen 08:30 Uhr und 16:45 Uhr berechnen:


- Startzeit: 08:30 Uhr
- Endzeit: 16:45 Uhr

Geben Sie die Start- und Endzeit in separate Zellen ein, zum Beispiel A2 und B2.

Subtrahieren Sie anschließend die Startzeit von der Endzeit. Erfassen Sie entsprechend in Zelle C2:

=B2-A2

Das Ergebnis für 16:45-08:30 sind 8:15 Stunden.



	A	B	C
1	Startzeit	Endzeit	Stunden
2	08:30	16:45	08:15
3			


Zeitdifferenz berechnen durch Subtraktion

Zeitformat der Zellen auf Uhrzeit stellen

Um das Ergebnis in Stunden und Minuten korrekt anzuzeigen, ändern Sie das **Zellenformat** auf **[hh]:mm**.

Falls das Ergebnis in der Zelle C2 nicht im Zeitformat dargestellt wird, dann gehen Sie wie folgt vor:

- Markieren Sie die entsprechende Zelle (C2); gegebenenfalls auch die anderen Zellen mit Zeitangaben, wenn diese nicht im Format hh:mm dargestellt sind.
- Drücken Sie die Tastenkombination **Strg + 1**.
- Es öffnet sich das Dialogfeld **Zellen formatieren**.
- Klicken Sie hier oben auf das Register **Zahlen**.
- Wählen Sie unter Kategorie den Eintrag **Benutzerdefiniert** aus.
- Erfassen Sie unter **Typ** das Format **[hh]:mm**.
- Bestätigen Sie das Format, indem Sie das Dialogfeld durch einen Klick auf **OK** schließen.



Zelle mit dem Format Uhrzeit hh:mm formatieren

Beispiel: Zeitdifferenz über zwei Tage und Mitternacht berechnen

Falls Ihre Zeitspanne über Mitternacht hinausgeht, wie zum Beispiel von 22:00 Uhr bis 06:00 Uhr am nächsten Tag, muss Excel erkennen, dass die Endzeit am Folgetag liegt. Hier ist eine Anpassung in der Formel notwendig.

Wieder werden die Uhrzeiten, deren Differenz ermittelt werden soll, in den Zellen A2 und B2 erfasst:

- A2: Startzeit 22:00 Uhr
- B2: Endzeit 06:00 Uhr des folgenden Tages

Verwenden Sie zur Berechnung der Stunden nun die folgende Formel in der Zelle C2:

=WENN(B2<A2; (B2+1)-A2; B2-A2)

Diese Formel prüft, ob die Endzeit (B2) kleiner ist als die Startzeit (A2). Ist dies der Fall, fügt Excel einen Tag (+1) zur Endzeit hinzu.

Das Ergebnis in Zelle C2 ist dann 8 Stunden (08:00). Ist die Startzeit kleiner als die Endzeit, dann gelten beide Zeiten für denselben Tag. In dem Fall wird die Startzeit von der Endzeit subtrahiert, wie im einfachen Beispiel oben.

C2 =WENN(B2<A2; (B2+1)-A2; B2-A2)					
	A	B	C	D	
1	Startzeit	Endzeit	Stunden		
2	22:00	06:00	08:00		
3					

Zeitunterschied berechnen über zwei Tage hinweg

Beispiel: Berechnung Dezimalstunden an einem Tag

In manchen Fällen ist es praktischer, das Ergebnis als Dezimalzahl darzustellen (zum Beispiel 8,25 Stunden statt 8:15), etwa für die Berechnung von Arbeitslöhnen. Auch für diesen Fall können Sie die Übernacht-Berechnung anwenden.

Um die Stunden als Dezimalwert anzuzeigen, multiplizieren Sie das Ergebnis der Subtraktion einfach mit 24. Die Formel in Zelle C2 lautet:

=(B2-A2)*24

Das Ergebnis für 16:45-08:30 ist: 8,25 Stunden.

C2 =(B2-A2)*24			
	A	B	C
1	Startzeit	Endzeit	Stunden
2	08:30	16:45	8,25
3			


Uhrzeitdifferenz im Dezimalformat berechnen

Zellformat auf Dezimalformat einstellen

Um das Ergebnis in Stunden und Dezimal-Minuten korrekt anzuzeigen, ändern Sie das Zellenformat auf **#.##0,00**.

Falls das Ergebnis in der Zelle C2 nicht in diesem Zeitformat dargestellt wird, dann gehen Sie wie folgt vor:

- Markieren Sie die entsprechende Zelle (C2).
- Drücken Sie die Tastenkombination **Strg + 1**.
- Es öffnet sich das Dialogfeld **Zellen formatieren**.
- Klicken Sie hier oben auf das Register **Zahlen**.
- Wählen Sie unter **Kategorie** den Eintrag **Zahl** aus.
- Wählen Sie **2 Dezimalstellen** aus und aktivieren Sie das Kontrollkästchen bei **1000er-Trennzeichen verwenden**.
- Bestätigen Sie das Format, indem Sie das Dialogfeld durch Klick auf **OK** schließen.



Zellformat auf Dezimalzahl einstellen

Beispiel: Dezimalzeit berechnen über Mitternacht und zwei Tage

Bei einer Berechnung über Mitternacht kann die gleiche Logik verwendet werden, wobei die Mitternacht-Formel eingesetzt wird:

- A2: Startzeit 22:00 Uhr
- B2: Endzeit 06:00 Uhr des folgenden Tages

Verwenden Sie zur Berechnung der Stunden nun die folgende Formel in der Zelle C2:

=WENN(B2<A2; (B2+1)-A2; B2-A2)*24

C2 X ✓ fx =WENN(B2<A2;(B2+1)-A2;B2-A2)*24					
	A	B	C	D	
1	Startzeit	Endzeit	Stunden		
2	22:00	06:00	8,00		
3					

Differenz zwischen zwei Uhrzeiten von aufeinanderfolgenden Tagen

Das Ergebnis lautet 8 Stunden in Dezimalform.

Hinweis: Stellen Sie sicher, dass das Zellenformat als Standard oder Zahl eingestellt ist, damit das Ergebnis als Dezimalwert angezeigt wird.

Zeitunterschied beträgt mehrere Tage

Wenn zwischen der Startzeit und der Endzeit nicht nur ein paar Stunden, sondern mehrere Tage liegen, dann können Sie die Zeitdifferenz in Stunden ebenfalls berechnen.

In diesem Fall müssen Sie außer der Uhrzeit auch noch das Datum für den Startzeitpunkt und für den Endzeitpunkt angeben – wie in der folgenden Abbildung.

	A	B	
1	Startdatum	Enddatum	
2	01.11.2024	06.11.2024	
3	Startzeit	Endzeit	
4	22:00	06:45	

Zeitangabe mit Datum und Uhrzeit

Die Anzahl der Stunden zwischen den beiden Tagen und Uhrzeiten berechnen Sie mit der folgenden Formel:

$=(B2+B4) - (A2+A4)$

In den Zellen B2 und A2 steht jeweils das Datum (Tag) und in den Zellen B4 und A4 die Endzeit und die Startzeit am angegebenen Datum.

C4	✕	✓	<i>fx</i>	$=(B2+B4) - (A2+A4)$
	A	B	C	
1	Startdatum	Enddatum		
2	01.11.2024	06.11.2024		
3	Startzeit	Endzeit	Stunden	
4	22:00	06:45	104:45	
5				

Zeitdifferenz zwischen mehreren Tagen

Schließlich können Sie die Ausgabe der Zeitdifferenz mit der folgenden Formel in Tage und Stunden anpassen.

$=\text{GANZZAHL}(C4)\&" \text{Tage und } "&\text{RUNDEN}(\text{REST}(C4;1)*24;2)\&" \text{Stunden}"$

D4 </						
---	--	--	--	--	--	--

Angabe der Zahl der Tage und Stunden einer Zeitdifferenz

Bearbeitungsdauer berechnen

Wie Sie mit einer Excel-Formel die Anzahl der Tage berechnen, die ein Ticket im IT-Ticket-System offen ist. Mit dieser Formel ermitteln Sie auch, wie lange an einem Arbeitspaket bereits gearbeitet wird. So können Sie den Zeitaufwand für das Projekt-Controlling ermitteln.

Zuletzt geändert am 11.03.2026



Warum wird die Bearbeitungsdauer berechnet?

In der IT-Abteilung und in der Service-Hotline ist das Ticket-System ein unverzichtbares Werkzeug, um Anfragen, Probleme und Aufgaben zu verwalten. Dabei ist wichtig zu wissen, welche Tickets offen sind und wie lange sie bereits offen sind oder waren.

Diese Information ist essenziell, um die Effizienz und Reaktionsfähigkeit des Supports zu überwachen. Eine zu lange Bearbeitungszeit kann auf interne Engpässe, überlastete Teams oder ineffiziente Prozesse hinweisen.

Deshalb sollte die Dauer, die ein Ticket im System verweilt, regelmäßig analysiert werden.

Diese Daten helfen nicht nur, die Kundenzufriedenheit zu verbessern, sondern auch, **wichtige KPIs (Key Performance Indicators)** wie die durchschnittliche Bearbeitungszeit oder die Anzahl unbearbeiteter Tickets im Blick zu behalten.

Warum Excel dafür nutzen?

Excel bietet eine einfache und flexible Möglichkeit, solche Analysen durchzuführen. Mit der richtigen Formel lässt sich schnell berechnen, wie viele Tage ein Ticket offen war – unabhängig davon, ob es bereits geschlossen wurde oder noch in Bearbeitung ist.

Außerdem können Sie mithilfe von Excel:

- Tickets mit langer Bearbeitungsdauer automatisch hervorheben,
- Durchschnittswerte und maximale Bearbeitungszeiten berechnen und
- die Ergebnisse in anschaulichen Diagrammen darstellen.

In diesem Beitrag erklären wir Ihnen eine praktische Excel-Formel, mit der Sie die offenen Tage für jedes Ticket in Ihrem System berechnen können.

Beispiel: Daten aus dem Ticket-System zur Analyse der Bearbeitungszeiten

In der folgenden Abbildung sehen Sie ein Beispiel mit Daten aus einem Ticket-System. Für

- jedes Ticket (Spalte A) sind
- das Datum der Eröffnung (Spalte B) und
- das Datum der Schließung, also der Tag, an dem das Ticket vollständig als bearbeitet markiert ist (Spalte C), angegeben.

Ist das Ticket noch nicht vollständig bearbeitet, bleibt der Eintrag in Spalte C leer.

Wurde in der Spalte C ein Schließungsdatum erfasst, dann soll in der Spalte D die **Dauer in Tagen** ausgegeben werden, wie lange das Ticket in Bearbeitung (offen) war.

Ist kein Schließungsdatum in der Spalte C erfasst, dann soll in der Spalte D die Anzahl der Tage dargestellt werden, wie lange das Ticket bereits in Bearbeitung (offen) ist.

	A	B	C	D
1	Ticket	Eröffnung	Schließung	Tage offen
2	1000	14.10.2024	14.10.2024	
3	1001	24.10.2024	26.10.2024	
4	1002	08.11.2024		
5	1003	10.11.2024		
6	1004	22.11.2024	23.11.2024	

Zeitangaben aus dem Ticket-System

Formel zur Berechnung der Zeitdauer

Hierfür können Sie die folgende Formel in Excel einsetzen:

=WENN(ISTLEER(C2); HEUTE()-B2; C2-B2)

Die Bezüge haben hierbei die folgende Bedeutung:

- **C2**: Hier befindet sich das Datum, wann das Ticket geschlossen wurde. Wurde das Ticket noch nicht geschlossen, dann ist die Zelle leer.
- **B2**: Hier befindet sich das Datum, an dem das Ticket eröffnet wurde.

D2					
	A	B	C	D	E
1	Ticket	Eröffnung	Schließung	Tage offen	
2	1000	14.10.2024	14.10.2024	0	
3	1001	24.10.2024	26.10.2024	2	
4	1002	08.11.2024		19	
5	1003	10.11.2024		17	
6	1004	22.11.2024	23.11.2024	1	

Formel zur Berechnung der Zeitdauer

Beachten Sie, dass mit dieser Formel der Tag der Eröffnung nicht mitgezählt wird. Entsprechend lautet das Ergebnis bei Eröffnung und Schließung am selben Tag 0. Zwischen dem 24.10.2024 und dem 26.10.2024 liegen zwei Tage.

Wollen Sie den Tag der Eröffnung und den Tag der Schließung mitzählen, ändern Sie die Formel zu:

=WENN(ISTLEER(C2); HEUTE()-B2; C2-B2) + 1

Woraus setzt sich die Formel zusammen?

Im Wahrheitstest der WENN-Funktion wird zunächst überprüft, ob die Zelle mit dem Schließungsdatum leer ist:

=WENN(ISTLEER(C2); HEUTE()-B2; C2-B2)

Ist die Zelle mit dem Schließungsdatum leer, dann wird durch die Funktion **HEUTE()** das Datum des aktuellen Tages ausgegeben und anschließend davon das Datum des Eröffnungstages subtrahiert:

=WENN(ISTLEER(C2); **HEUTE()-B2**; C2-B2)

Sie erhalten so die Anzahl der Tage, wie lange das Ticket bereits offen, also noch nicht vollständig bearbeitet wurde.

Hinweis: Die Funktion **HEUTE()** wird automatisch aktualisiert, wenn die Arbeitsmappe neu berechnet oder geöffnet wird.

Wurde ein Schließungsdatum in der Zelle C2 erfasst, dann wird vom Wert des Schließungsdatums in der Zelle C2 der Wert des Eröffnungsdatums in der Zelle B2 subtrahiert:

=WENN(ISTLEER(C2); HEUTE()-B2; **C2-B2**)

Sie erhalten so die Bearbeitungszeit des Tickets in Tagen.

Formel zur Berechnung der Zeitdauer mit Stunden und Minuten

Wollen Sie die Bearbeitungszeit nicht nur in Tagen, sondern exakt in Tagen unter Berücksichtigung der tatsächlichen Stunden und Minuten berücksichtigen, dann erfassen Sie in den Spalten B und C nicht nur das Datum, sondern zusätzlich die entsprechende Uhrzeit.

Anschließend ersetzen Sie in der Formel die Funktion HEUTE() durch die Funktion **JETZT()**.

=WENN(ISTLEER(C2); **JETZT()-B2**; C2-B2)

Die JETZT()-Funktion in Excel gibt das aktuelle Datum **und die aktuelle Uhrzeit** basierend auf der Systemuhr des Computers zurück. Sie erhalten so bei offenen Tickets die aktuelle Bearbeitungszeit unter Berücksichtigung der aktuellen Uhrzeit.

Hinweis: Die Funktion JETZT() wird automatisch aktualisiert, wenn die Arbeitsmappe neu berechnet oder geöffnet wird.

D2				=WENN(ISTLEER(C2);JETZT()-B2;C2-B2)
	A	B	C	D
1	Ticket	Eröffnung	Schließung	Tage offen
2	1000	14.10.2024 08:00	14.10.2024 10:00	0,083
3	1001	24.10.2024 11:00	26.10.2024 08:00	1,875
4	1002	8.11.2024 10:00		19,064
5	1003	10.11.2024 12:00		16,981
6	1004	22.11.2024 15:00	23.11.2024 09:00	0,750

Formel zur Berechnung der Zeitdauer mit Stunden und Minuten

Die Zeitdauer in Tagen wird als Dezimalzahl angegeben. Der Nachkommateil entspricht dem Bruchteil des Tages, also der Zahl der Stunden. Um das besser sichtbar zu machen, können Sie das Ergebnis mit folgender Formel aufbereiten:

=GANZZAHL(D2)&" Tage und "&RUNDEN(REST(D2; 1) * 24; 2)&" Stunden"

E2					=GANZZAHL(D2)&" Tage und "&RUNDEN(REST(D2;1)*24;2)&" Stunden"
	A	B	C	D	E
1	Ticket	Eröffnung	Schließung		Tage offen
2	1000	14.10.2024 08:00	14.10.2024 10:00	0,083	0 Tage und 2 Stunden
3	1001	24.10.2024 11:00	26.10.2024 08:00	1,875	1 Tage und 21 Stunden
4	1002	8.11.2024 10:00		19,070	19 Tage und 1,68 Stunden
5	1003	10.11.2024 12:00		16,987	16 Tage und 23,68 Stunden
6	1004	22.11.2024 15:00	23.11.2024 09:00	0,750	0 Tage und 18 Stunden

Zeitdauer in der Aufbereitung als Tage und Stunden

Wollen Sie bei Ihrer Darstellung der Bearbeitungsdauer die Angabe „0 Tage“ weglassen, und soll aus „1 Tage“ das korrekte „1 Tag“ werden, nutzen Sie diese Formel:

=WENNS(D2<1; RUNDEN(REST(D2; 1) * 24; 2)&" Stunden"; D2<2; GANZZAHL(D2)&" Tag und "&RUNDEN(REST(D2; 1) * 24; 2)&" Stunden"; D2>=2; GANZZAHL(D2)&" Tage und "&RUNDEN(REST(D2; 1) * 24; 2)&" Stunden")

Mit der Funktion **WENNS()** wird geprüft, wie viele Tage ein Ticket in Bearbeitung war. Entsprechend wird die richtige Bezeichnung ausgewählt.

(Wir danken einem aufmerksamen Leser unseres Newsletters für diesen Excel-Tipp.)

F2					
=WENNS(D2<1;RUNDEN(REST(D2;1)*24;2)&" Stunden";D2<2;GANZZAHL(D2)&" Tag und "&RUNDEN(REST(D2;1)*24;2)&" Stunden";D2>=2;GANZZAHL(D2)&" Tage und "&RUNDEN(REST(D2;1)*24;2)&" Stunden")					
	A	B	C	D	E
1	Ticket	Eröffnung	Schließung		Tage offen
2	1000	14.10.2024 08:00	14.10.2024 10:00	0,083	0 Tage und 2 Stunden
3	1001	24.10.2024 11:00	26.10.2024 08:00	1,875	1 Tage und 21 Stunden
4	1002	16.2.2025 06:00		19,107	19 Tage und 2,57 Stunden
5	1003	18.2.2025 12:00		16,857	16 Tage und 20,57 Stunden
6	1004	22.11.2024 15:00	23.11.2024 09:00	0,750	0 Tage und 18 Stunden

Aufbereitete Darstellung der Ticketdauer

Mit denselben Formeln können Sie auch jede andere **Zeitdauer** berechnen und anschließend auswerten. Das kann zum Beispiel im **Controlling** helfen, wenn Sie beispielsweise wissen wollen,

- wie lange ein Projekt-Arbeitspaket bereits bearbeitet wird,
- wie lange die Wartung einer Maschine gedauert hat oder
- wie lange ein Fahrzeug unterwegs war.

Es gibt viele solcher Fälle, in denen Sie die genaue Dauer einer Tätigkeit oder eines Prozesses berechnen wollen. Die Formeln sind immer dieselben, wie sie oben erläutert sind.

Zeitangaben in Excel auf die Viertelstunde runden

So runden Sie eine Zeitangabe auf das nächste 15-Minuten-Intervall in Excel. Aus 11:55 Uhr wird dann automatisch 12:00 Uhr. Zudem erfahren Sie, wie Sie mit einer einfachen Funktion Zeitangaben immer aufrunden oder abrunden.

Zuletzt geändert am 11.03.2026



Warum eine Runden-Funktion für Zeit?

Das Runden auf 15-Minuten-Intervalle ist in vielen Arbeitsbereichen hilfreich, sei es bei der Arbeitszeiterfassung, der Terminplanung oder der Datenanalyse.

Die Rundung auf ein festgelegtes Intervall kann für verschiedene Zwecke nützlich sein:

- Zeiten für die **Lohnabrechnung** vereinheitlichen: Zeitangaben exakt auf Viertelstunden zu runden, bietet eine klare Grundlage für die Abrechnung.
- Termine und **Meetings** planen: Runden Sie geplante Zeiten auf ein 15-Minuten-Intervall, um Kalender übersichtlicher zu gestalten.
- **Logbuch- oder Messdaten** aggregieren: Zeitangaben auf feste Intervalle zu runden, erleichtert die Analyse und das Erkennen von Mustern.


Die Formel für das 15-Minuten-Intervall

In Excel können Sie die Funktion **VRUNDEN()** verwenden, um eine Uhrzeit auf das nächste Viertelstunden-Intervall zu runden. Die entsprechende Formel lautet:

=VRUNDEN(A2; "0:15")

- **A2**: Die Formel geht in diesem Beispiel davon aus, dass die zu rundende Uhrzeit in der Zelle A2 steht.
- **"0:15"**: Das Intervall, auf das gerundet werden soll – in diesem Fall 15 Minuten.

Angenommen, in der Zelle A2 steht die Uhrzeit 12:07. Wenn Sie die Formel `=VRUNDEN(A2; "0:15")` eingeben, wird die Zeit auf 12:00 abgerundet. Bei der Uhrzeit 12:08 rundet die Formel `=VRUNDEN(A2; "0:15")` auf 12:15 auf.




	A	B	C	D
1	Zeiten	gerundet		
2	12:07	12:00		
3	12:08	12:15		
4	12:20	12:15		
5	12:40	12:45		

Zeitangabe runden mit VRUNDEN()

Ergebniszellen als Uhrzeit formatieren

Damit die gerundeten Zeiten als Uhrzeit dargestellt werden, müssen Sie die entsprechenden Zellen mit der Rundungsfunktion als **Uhrzeit** formatieren. So gehen Sie vor:

- Markieren Sie die entsprechenden Zellen.
- Drücken Sie die Tastenkombination **Strg + 1**.
- Es öffnet sich das Dialogfeld **Zellen formatieren**.
- Klicken Sie auf das Register **Zahlen**.
- Wählen Sie unter **Kategorie** den Eintrag **Uhrzeit** aus.
- Unter **Typ** wählen Sie das gewünschte **Uhrzeitformat** aus.
- Bestätigen Sie Ihre Einstellung, indem Sie das Dialogfeld durch einen Klick auf **OK** schließen.



Zellformat auf Uhrzeit einstellen

So können Sie immer auf die nächsten 15 Minuten aufrunden

Falls Sie die Zeit **immer** auf das nächste höhere 15-Minuten-Intervall **aufrunden** möchten, verwenden Sie die Funktion:

=OBERGRENZE(A2; "0:15")

- **A2**: Die Formel geht in diesem Beispiel davon aus, dass die zu rundende Uhrzeit in der Zelle A2 steht.
- **"0:15"**: Das Intervall, auf das **aufgerundet** werden soll – in diesem Fall 15 Minuten.

B2 X ✓ <i>fx</i> =OBERGRENZE(A2;"0:15")				
	A	B	C	D
1	Zeiten	gerundet		
2	12:07	12:15		
3	12:08	12:15		
4	12:20	12:30		
5	12:40	12:45		

Mit OBERGRENZE() Zeitangabe immer aufrunden

So können Sie immer auf die nächsten 15 Minuten abrunden

Falls Sie die Zeit immer auf das vorherige 15-Minuten-Intervall **abrunden** möchten, verwenden Sie die Funktion:

=UNTERGRENZE(A2; "0:15")

- **A2**: Die Formel geht in diesem Beispiel davon aus, dass die zu rundende Uhrzeit in der Zelle A2 steht.
- **"0:15"**: Das Intervall, auf das **abgerundet** werden soll – in diesem Fall 15 Minuten.

B2 X ✓ <i>fx</i> =UNTERGRENZE(A2;"0:15")				
	A	B	C	D
1	Zeiten	gerundet		
2	12:07	12:00		
3	12:08	12:00		
4	12:20	12:15		
5	12:40	12:30		

Mit UNTERGRENZE() Zeitangabe immer abrunden

Zeitangabe auf halbe Stunde runden

Die drei Funktionen VRUNDEN(), OBERGRENZE() und UNTERGRENZE() funktionieren natürlich auch für andere Rundungen. Wenn Sie eine Zeitangabe auf die halbe Stunde runden, aufrunden oder abrunden wollen, ersetzen Sie in den Formeln einfach den Parameter "**0:15**" durch "**0:30**".

Auch hier müssen Sie der Ergebniszelle das **Format Uhrzeit** zuweisen.

Datum- und Zeitformate in Excel selbst definieren

Datum und Uhrzeit können in Excel in unterschiedlichen Formaten angezeigt werden. Mit den Einstellungen in den benutzerdefinierten Zahlenformaten legen Sie beispielsweise fest, ob 1.1.22 oder 1. Januar 2022 angezeigt wird.

Zuletzt geändert am 11.03.2026




Wenn Sie ein Datum oder eine Uhrzeit in einer Zelle erfassen, dann werden diese Werte in einem Standardformat für Datum oder Uhrzeit angezeigt.

Auch für Datums- und Uhrzeitwerte können Sie **benutzerdefinierte Zahlenformate** erstellen und so Ihre eigenen Darstellungen definieren, wie die Zeitwerte in Excel dargestellt werden sollen.

Datum und Uhrzeit im Standardformat eingeben

An folgendem Beispiel erkennen Sie, welche Einstellungen und Formatierungen Sie bei Datum und Uhrzeit mit Excel vornehmen können. Erfassen Sie in einer Zelle ein Datum wie zum Beispiel 1.1.22 und drücken Sie die Entertaste.



Datum 1.1.22 wird als 01.01.2022 dargestellt

Obwohl Sie die Jahreszahl nur mit zwei Ziffern erfasst haben, wird das Datum mit vier Jahresziffern in der Zelle angezeigt. Excel erkennt bei der Eingabe automatisch, dass es sich um einen Datumswert handelt, und weist diesem Wert ein **Standardformat für ein Datum (TT.MM.JJJJ)** zu.

Dieses **Standardformat** basiert auf den **regionalen Einstellungen für Datum und Uhrzeit**, die in der **Systemsteuerung** angegeben sind. In diesem Beispiel ist das Standardformat so formatiert, dass die Jahreszahl grundsätzlich mit vier Ziffern dargestellt wird.


Hinweis: Hätten Sie eine Uhrzeit in der Zelle erfasst, wäre Excel analog vorgegangen und hätte ein **Standarduhrzeitformat (hh:mm)** als Zahlenformat hinterlegt. Wenn Sie beispielsweise 8:33 als Uhrzeit in einer Zelle eintragen, wird automatisch angezeigt: 08:33

Datumswerte in Excel definieren

Wie werden Datumswerte und Uhrzeiten in Excel formatiert? Und lässt sich die Darstellung anpassen?

Dazu wird das benutzerdefinierte Zahlenformat genauer betrachtet:

- Markieren Sie die Zelle, die den Datumswert beinhaltet.
- Drücken Sie die Tastenkombination **Strg + 1**.
- Es öffnet sich das Dialogfeld **Zellen formatieren**.
- Aktivieren Sie das Register **Zahlen**.
- Wählen Sie am linken Rand die Kategorie **Benutzerdefiniert** aus.
- Sie bekommen jetzt im rechten Bereich unter **Typ** den Format-Code für das dargestellte Datum angezeigt: **TT.MM.JJJJ**



Formatsymbole für das angezeigte Datum TT.MM.JJJJ

Die hier verwendeten Format-Codes haben die folgende Bedeutung:

- T = Tag
- M = Monat
- J = Jahr

Durch die sinnvolle Anordnung oder Wiederholung der Format-Codes können Sie hierbei eigene sinnvolle und nützliche Datumsanzeigen definieren.

Insbesondere bei der Tages- und Monatsangabe können hier wichtige Informationen wie Tages- und Monatsnamen automatisch ermittelt und dargestellt werden:

- **T** Tage als **1-31**
- **TT** Tage als **01-31**
- **TTT** Tage als **So** bis **Sa**
- **TTTT** Tage als **Sonntag** bis **Samstag**
- **M** Monate als **1-12**
- **MM** Monate als **01-12**
- **MMM** Monate als **Jan** bis **Dez**
- **MMMM** Monate als **Januar** bis **Dezember**
- **MMMMM** Monate mit dem **ersten Buchstaben des Monats**
- **JJ** Jahre als **00** bis **99**
- **JJJJ** Jahre als **1900** bis **9999**

In der folgenden Abbildung bekommen Sie einige Beispiele zu den Format-Codes dargestellt.

	A	B	C
1	Datum	Zahlenformat	Anzeige
2	01.12.2022	TT.MM.JJJJ	01.12.2022
3	01.12.2022	TT.MM.JJ	01.12.22
4	01.12.2022	T.M.JJJJ	1.12.2022
5	01.12.2022	T	1
6	01.12.2022	TT	01
7	01.12.2022	TTT	Do
8	01.12.2022	TTTT	Donnerstag
9	01.12.2022	M	12
10	01.12.2022	MM	12
11	01.12.2022	MMM	Dez
12	01.12.2022	MMMM	Dezember
13	01.12.2022	MMMMM	D
14	01.12.2022	JJ	22
15	01.12.2022	JJJJ	2022
16	01.12.2022	TT.MMM.JJJJ	01.Dez.2022
17			

Beispiele für benutzerdefinierte Datumsformate

Uhrzeiten in Excel formatieren

Analog zu den Format-Codes für Datumswerte gibt es auch entsprechende Codes für die Uhrzeiten:

- h = Stunde
- m = Minute
- s = Sekunde

Auch hier können Sie durch die Anordnung/Wiederholung der Format-Codes unterschiedliche Darstellung erzeugen:

- **h** Stunden als bis **23**
- **hh** Stunden als **00** bis **23**
- **m** Minuten als bis **59**
- **mm** Minuten als **00** bis **59**
- **s** Sekunden als bis **59**
- **ss** Sekunden als **00** bis **59**

	A	B	C
1	Uhrzeit	Zahlenformat	Anzeige
2	08:05:15	hh:mm:ss	08:05:15
3	08:05:15	hh:mm	08:05
4	08:05:15	h:m:s	8:5:15
5	08:05:15	hh:ss	08:15
6	08:05:15	mm:ss	05:15
7	08:05:15	h	8
8	08:05:15	hh	08
9	08:05:15	m	5
10	08:05:15	mm	05
11	08:05:15	s	15
12	08:05:15	ss	15
13			

Beispiele für benutzerdefinierte Uhrzeitformate

Negative Uhrzeiten in Excel darstellen

Ist die Differenz zwischen zwei Uhrzeiten im Format hh:mm negativ, kann Excel das Ergebnis im Standard-Datumsystem nicht anzeigen. Erst wenn das Datumsystem geändert wird, sehen Sie den negativen Wert für die Uhrzeit. Wie funktioniert das?

Zuletzt geändert am 11.03.2026



Wie Excel negative Uhrzeiten oder Zeitdifferenzen im Format hh:mm darstellt

In Excel werden keine negativen Uhrzeiten dargestellt. In der folgenden Abbildung werden die Abweichungen zwischen Plan- und Ist-Zeiten ermittelt, indem sie in der Spalte C subtrahiert werden: Ist-Zeit minus Plan-Zeit.

Ist das Ergebnis der Subtraktion positiv, kann der berechnete Wert problemlos im **Format hh:mm** dargestellt werden. Bei negativen Ergebnissen sehen Sie in der Zelle Rautenzeichen das **#-Zeichen**.

	A	B	C
1	Plan	Ist	Differenz
2	10:00	12:00	02:00
3	08:00	12:00	04:00
4	12:00	10:00	#####
5			

Negative Uhrzeit oder Zeitdifferenzen werden in Excel nicht dargestellt

Es handelt sich hierbei nur um ein **Darstellungsproblem**, das beim Zahlenformat mit **Uhrzeiten** auftritt. Ändern Sie das Zahlenformat in der Spalte C in das Zahlenformat

Standard, dann erkennen Sie, dass es sich um keinen Fehler handelt.

Markieren Sie hierzu die Zellen C2:C4 und drücken Sie anschließend die Tastenkombination **Strg + 1**. Es öffnet sich das Dialogfenster **Zellen formatieren**.

Wählen Sie hier im Register **Zahlen** die Kategorie **Standard** aus.

Bestätigen Sie Ihre Einstellung, indem Sie auf **OK** klicken. Die folgende Abbildung zeigt das Ergebnis.

	A	B	C
1	Plan	Ist	Differenz
2	10:00	12:00	0,08333333
3	08:00	12:00	0,16666667
4	12:00	10:00	-0,08333333
5			

Im Zahlenformat Standard werden auch negative Uhrzeiten oder Zeitdifferenzen dargestellt

Durch die Darstellung der Werte im Zahlenformat **Standard** ist ersichtlich, dass Excel in der Zelle C4 sehr wohl einen negativen Wert berechnet hat.

Da es grundsätzlich keine negativen Uhrzeiten gibt, werden diese im Zahlenformat **Uhrzeit** aber nicht als negative Werte dargestellt. Es werden lediglich **#-Zeichen** in der Zelle anstelle von negativen Uhrzeiten angezeigt.

Das Nichtanzeigen von negativen Uhrzeiten in Excel beruht auf dem **1900-Datumsystem**, das standardmäßig in Excel aktiviert ist. Ändern Sie das Datumsystem auf **1904-Datumswerte** ab, dann können Sie negative Uhrzeiten auch in Ihren Excel-Tabellen darstellen.


Wie Sie negative Uhrzeiten oder Zeitdifferenzen in Excel anzeigen

Öffnen Sie hierzu die Excel-Optionen, indem Sie im Menüband die Befehlsfolge Registerkarte **Datei** > Befehl **Optionen** anklicken. Es öffnet sich das Dialogfeld **Excel-Optionen**.

Aktivieren Sie hier am linken Rand die Kategorie **Erweitert**.

Aktivieren Sie in der Gruppe **Beim Berechnen dieser Arbeitsmappe** das Kontrollkästchen **1904-Datumswerte verwenden**.

Bestätigen Sie Ihre Einstellung, indem Sie die Excel-Optionen durch einen Klick auf **OK** schließen.



1904-Datumswerte in Excel einstellen und verwenden

Durch die Verwendung der 1904-Datumswerte in der Arbeitsmappe werden jetzt auch **negative Uhrzeiten** in den Zellen im Zahlenformat **Uhrzeit** (hh:mm) dargestellt.

Wichtig: Beachten Sie, dass die Einstellung der 1904-Datumswerte für die komplette Arbeitsmappe gilt.

	A	B	C
1	Plan	Ist	Differenz
2	10:00	12:00	02:00
3	08:00	12:00	04:00
4	12:00	10:00	-02:00
5			

Negative Uhrzeit wird im 1904-Datumsformat richtig dargestellt

Seien Sie sich bewusst, dass die Aktivierung der 1904-Datumswerte auch **Risiken** beinhaltet und zu **Problemen** führen kann. Durch die Aktivierung dieser Option wird der Beginn der Excel-Zeitrechnung in der Arbeitsmappe auf den 01.01.1904 gelegt. Dieses Datum entspricht dann der seriellen Zahl 0, der 02.01.1904 der Zahl 1 usw.

Beim Standard für das Datumsystem entspricht die serielle Zahl 1 dem Datum 01.01.1900. Aktivieren Sie jetzt das Datumsystem 1904 in den Excel-Optionen, dann werden **bereits erfasste Datumswerte** automatisch um 4 Jahre und 1 Tag erhöht.

Dies geschieht ohne Warnung! Sie sollten sich bewusst sein, dass sich in Ihrer Excel-Datei alle eingetragenen Datumsangaben entsprechend ändern!

Hintergrund: Datumsysteme in Excel

Im Menü **Datei > Optionen > Erweitert** kann für eine Excel-Datei das Datumsystem festgelegt werden. Es gibt zwei Datumsysteme:

- Standard **1900-Datumsystem**: Die Zählung beginnt mit dem 01.01.1900 und dem Zahlenwert 1.
- **1904-Datumsystem**: Die Zählung beginnt mit dem 02.01.1904 und dem Zahlenwert 1. Der 01.01.1904 entspricht dem Zahlenwert 0.

Alle bestehenden Datumsangaben in einer Excel-Datei werden beim Umschalten automatisch um 4 Jahre und 1 Tag erhöht bzw. verringert.

Hinweis

Negative Uhrzeiten im 1900-Datumsystem

Sie können eine negative Uhrzeit auch im 1900-Datumsystem darstellen. Allerdings braucht es dazu eine Formel, die wir Ihnen in diesem Excel-Tipp vorstellen und erklären: **Negative Uhrzeiten im 1900-Datumsystem darstellen.**

Negative Uhrzeiten im 1900-Datumsystem darstellen

Excel kennt in seinem Standard-Datumsystem keine negativen Uhrzeiten. Wenn Sie dennoch eine negative Zeit darstellen wollen, brauchen Sie eine spezielle Formel. Und wenn Sie mit einer negativen Zeitangabe rechnen müssen, hilft die Erweiterung zur Matrixformel.

Zuletzt geändert am 11.03.2026



Negative Zeitangaben resultieren aus Berechnungen mit Uhrzeiten

Wenn Sie in Excel eine negative Uhrzeit in einer Zelle erfassen, dann wird diese standardmäßig mit einem Gartenzaun oder Rautenzeichen (**#**) dargestellt. Dies ist in der Praxis oft ein Problem, da es viele Fälle gibt, in denen Sie mit negativen Uhrzeiten arbeiten müssen.

Vergleichen Sie zum Beispiel bei Mitarbeitenden die geleistete Ist-Arbeitszeit mit der täglichen Soll-Arbeitszeit, dann können sich auch negative Abweichungen ergeben (siehe folgende Abbildung).

C2					=A2-B2				
	A	B	C	D					
1	Ist	Plan	Differenz						
2	10:00	08:00	02:00						
3	10:00	08:00	02:00						
4	06:00	08:00	#####						
5	10:00	08:00	02:00						
6	06:00	08:00	#####						
7									

Durch Berechnung erhalten Sie eine negative Zeitangabe oder Uhrzeit

Problem: Excel kennt keine negative Uhrzeit im 1900-Datumsystem

In den Zellen mit negativen Abweichungen zeigt Excel den Gartenzaun anstelle einer negativen Uhrzeit an. Im üblichen 1900-Datumsystem von Excel lassen sich keine negativen Uhrzeiten in einer Zelle angeben.

Eine Eingabe oder eine Berechnung mit dem Ergebnis **-hh:mm** ist also leider nicht möglich.

Lösung: Formel zur Darstellung negativer Uhrzeiten in Excel

Da Excel keine negativen Uhrzeiten in einer Zelle darstellen kann, müssen Sie Excel mit der folgenden Formel überlisten:

=WENN(A2-B2<0; TEXT(ABS(A2-B2); "-hh:mm"); A2-B2)

Die Formel geht dabei von den folgenden Annahmen aus (siehe Abbildung oben):

- **A2**: Zelle, welche die Uhrzeit 1 (Ist) beinhaltet
- **B2**: Zelle, welche die Uhrzeit 2 (Plan) beinhaltet
- Uhrzeit 2 wird von Uhrzeit 1 abgezogen, was zu einem negativen Ergebnis führen kann.

Die Formel geht dabei wie folgt vor: Sie überprüft zunächst, ob es sich um eine positive oder negative Zeitdifferenz handelt:

=WENN(A2-B2<0; TEXT(ABS(A2-B2); "-hh:mm"); A2-B2)

Ist die Differenz negativ (<0), dann wird wie folgt vorgegangen:

=WENN(A2-B2<0; TEXT(ABS(A2-B2); "-hh:mm"); A2-B2)

Das bedeutet:

- Es wird die absolute Zahl der Differenz mit der Funktion **ABS()** berechnet; das heißt, das negative Vorzeichen bei der Differenz wird nicht beachtet.
- Das Ergebnis wird anschließend mit der Funktion **TEXT()** in einen Text umgewandelt und mit einem Uhrzeitzahlenformat formatiert, dem ein Minuszeichen vorangestellt wird; dies geschieht über den Parameter **"-hh:mm"**.
- Sie bekommen somit die negative Uhrzeit als Text und nicht als Zahl in der Zelle dargestellt.

Ist die Differenz A2-B2 positiv, wird die Differenz ohne weitere Umformung berechnet:

=WENN(A2-B2<0; TEXT(ABS(A2-B2); "-hh:mm"); A2-B2)

C2		=WENN(A2-B2<0;TEXT(ABS(A2-B2);"-hh:mm");A2-B2)					
	A	B	C	D	E	F	G
1	Ist	Plan	Differenz				
2	10:00	08:00	02:00				
3	10:00	08:00	02:00				
4	06:00	08:00	-02:00				
5	10:00	08:00	02:00				
6	06:00	08:00	-02:00				
7							

Darstellung von negativen Zeitangaben in Excel

Problem: Mit negativen Uhrzeiten im Textformat lässt sich nicht rechnen

Wollen Sie jetzt über alle Differenzen (Spalte C) eine Summe bilden, dann müssen Sie aufpassen, da die negativen Uhrzeiten keine Zahlen, sondern Texte in Excel darstellen.

Um dies genauer zu verdeutlichen, wird im Beispiel aus der vorigen Abbildung eine Summe mit der Funktion **SUMME()** gebildet.

	A	B	C	D
1	Ist	Plan	Differenz	
2	10:00	08:00	02:00	
3	10:00	08:00	02:00	
4	06:00	08:00	-02:00	
5	10:00	08:00	02:00	
6	06:00	08:00	-02:00	
7			06:00	
8				

Rechnen mit negativen Uhrzeiten kann zu falschen Ergebnissen führen

Das Ergebnis sieht auf den ersten Blick gut aus, ist aber falsch, da das richtige Ergebnis 2:00 lauten müsste. Excel hat die negativen Uhrzeiten bei der Summenbildung nicht berücksichtigt, da diese Texte und keine Zahlen darstellen.

Lösung: Formel für das Rechnen mit negativen Uhrzeiten zur Matrixformel erweitern

Sie müssen für diese Aufgabenstellung wieder eine spezielle Summenformel erstellen:

=WENN(SUMME(A2:A6-B2:B6)<0; TEXT(ABS(SUMME(A2:A6-B2:B6)); "-hh:mm"); SUMME(A2:A6-B2:B6))

Es handelt sich dabei um die Formel von oben, die zu einer **Matrixformel** erweitert ist. Damit können Sie die entsprechenden Zeilendifferenzen – egal ob Zahl oder Text – summieren.

Mit dieser Formel wird zunächst für jede einzelne Zeile die Differenz ermittelt und anschließend aufsummiert. Da die dargestellten Uhrzeiten von Excel wie Dezimalzahlen behandelt werden, ist das möglich.

Die summierte Zahl wird dann wieder in einen Text umgewandelt, sodass auch negative Summen als Uhrzeit dargestellt werden können.

Wichtig: Da es sich um eine Matrixformel handelt, bestätigen Sie die Formel nicht mit **Enter**, sondern mit der Tastenkombination **Strg + Umschalttaste + Enter**. Die Formel wird dann automatisch durch geschweifte Klammern **{ }** eingerahmt.

C7

Mit der Matrixformel ist das Rechenergebnis mit negativen Uhrzeiten korrekt

Empfehlungen aus dem Management-Handbuch

Mit Excel-Datum rechnen

Wie Sie eine Datumsangabe in Excel nutzen, um die Datenanalyse zu verbessern und wichtige Informationen herausfiltern wie die Anzahl der Arbeitstage, die Umstellung von Sommer- auf Winterzeit, die korrekte Kalenderwoche oder das Quartal eines Datums. Wichtige Informationen für jedes Reporting.

<https://www.business-wissen.de/id/kapitel/310/>

Nutzen Sie als
Premium-Mitglied
alle
Handbuch-Kapitel
mit mehr als
3.000 Checklisten und Excel-Vorlagen

Jetzt anmelden

www.business-wissen.de/anmelden/

Impressum

b-WISE GmbH Business Wissen Information Service
Bismarckstraße 21
76133 Karlsruhe
DEUTSCHLAND

service@business-wissen.de

Telefon +49 721 18397-0

Copyright 2026, b-wise GmbH, All Rights Reserved